

## Képletek

*Olyan egyenletek, amelyek a munkalapon szereplő értékekkel számításokat hajtanak végre.*

- **Jellemzői:**
  - egyenlőségjellel = kezdődik
- **Képlet részei:**
  1. Számítási műveleti jelek.
  2. Állandók.
  3. Hivatkozások.
  4. Operátorok.
  5. Függvények.

### 1. Számítási műveleti jelek

+ összeadás pl. =6+3

- kivonás (és ellentett képzése pl. -1) pl. =6-2

\* szorzás pl. =6\*3

/ osztás pl. =6/3

% százalék pl. =20%

^ hatványozás pl.=3^2 (matematikailag: 3<sup>2</sup>)

**Művelek sorrendje – mint a matematikában.**

- balról jobbra haladva,
- először a szorzást, osztást,
- majd az összeadást, kivonást.

Pl. =5+2\*3 Végrehajtás: 2\*3=6, majd 5+6

**Zárójel használata – mint a matematikában.**

Először a zárójelben lévő műveletet kell elvégezni.

Pl. =(5+2)\*3 Végrehajtás: 5+2=7, majd 7\*3

Pl. =(B4+30)/SZUM(D5:F5)

Végrehajtás; B4-es cellában szereplő értékhez hozzá adunk 30-at. Majd D5, F5 cellák tartalma, és a közük eső cellák tartalmának összege. Majd az osztás elvégzése.

### 2. Állandók

*Olyan értékek, amit nem kell kiszámítani.*

Állandó lehet:

- **szöveg**  
A képletben szereplő szöveget idéző jelek közé kell írni.  
Pl. "szöveg"  
="Negyedévi bevétel"& SZUM(B2:D21)

↑  
1 érték: állandó

↑  
Két értéket összefűző  
operátor &

←  
2 érték: Összeadó – szumma függvény,  
amelyben a számított értékek változók,  
mert a cellára hivatkozik.

- **szám**  
A képletben nem cellahivatkozást, hanem állandó értékeket (számokat) használunk.  
Pl. =30+20+100

↑ ↑ ↑  
Állandó értékek a képletben.

### **3. Hivatkozások**

A hivatkozás rámutat az adat helyére.

**Jellemzői:**

- A hivatkozás a **munkalap celláját**, vagy celláit **azonosítja**.
- Meghatározza, hogy a képletben használni kívánt adat, adatok hol találhatóak.
- Az adatok **változóként** szerepelnek a képletben, ha hivatkozással mutatunk rájuk.

**Módja – azaz mire hivatkozhatunk:**

- Egy cellára. Pl. A1
- Több összefüggő cellára = tartomány: pl. A1:B10
  - Több összefüggő cellát tartománynak nevezünk.
  - A tartományok téglalap alakúak.
  - A tartománynak megadjuk a kezdő és záró celláját.
  - A tartománynak kezdő és záró celláját kettőspont választja el (:).
- Egész sorra, sorokra.
  - Egy egész sor össze cellájára, a sor számával. pl. 5:5 (ötödik sor)
  - Több egymás melletti sor összes cellájára az első és az utolsó sor számával. pl. 5:10 (5-10.sor)
- Egész oszlopra, oszlopokra.
  - Egy egész oszlop össze cellájára, az oszlop betűjelével. Pl. B:B (B oszlop)
  - Több egymás melletti oszlop összes cellájára az első oszlop és az utolsó oszlop betűjelével. pl. B:E (B-E oszlopok)

**Típusai (fajtái):**

- **Relatív hivatkozás:**
  - Ha a képletet tartalmazó cella helye változik (pl. másoljuk), a hivatkozás is módosul.
  - 1. példa:
    - Pl. C3 cellába összeadjuk A3 és B3 cella tartalmát a következő képlettel; =A3+B3.
    - Ha a képletet lefele másoljuk, a C4 cellába a képlet a következőre módosul; =A4+B4.
    - A képletet lefele és fölfelé másolva a képletben a sorok száma változik. Lefele másolva nő a sorok száma, felfelé másolva csökken.
  - 2. példa:
    - Pl. C4-es cellában összeadjuk a C2 és C3 cella tartalmát a következő képlettel: =C2+C3.
    - Ha a képletet jobbra másoljuk, a D4 cellában a képlet a következőre módosul; =D2+D3
    - A képletet jobbra és balra másolva a képletben az oszlopok száma változik. Jobbra másolva növekszi az angol ábécé szerint az oszlopok neve, balra másolva csökken.
- **Abszolút hivatkozás:**
  - A képletben a hivatkozás mindig az adott helyre mutat.
  - Ha képletet tartalmazó cella helye megváltozik (pl. másoljuk), a hivatkozás nem módosul.

- Az abszolút hivatkozást a **cella rögzítésére használt operátor \$ (dollár jel)** teszi lehetővé. A jelet betesszük a cella helyét jelző oszlop és sor elé is. Pl. A1 cella rögzítése; =\$A\$1 Képletben; =B2\*\$A\$1 Ha ezt a képletet lefele másoljuk; =B3\*\$A\$1, =B4\*\$A\$1, stb.

- **Vegyes hivatkozás**

- A cellát azonosító oszlop és sor közül az egyik relatív, a másik abszolút hivatkozású. Ami előtt \$ jel van az abszolút, ami előtt nincs, az relatív.  
=B4\*A\$1, =B4\*\$A1

1. **Abszolút oszlop és relatív sor**

A képletben az oszlop helye állandó (rögzített). Másoláskor az oszlop (ami rögzített) nem változik a képletben. Csak a sor változik.

Pl. jobbra másoláskor =B4\*\$A1 → =C4\*\$A1

Lefele másoláskor: =B4\*\$A1 → =B5\*\$A1

2. **Relatív oszlop és abszolút sor**

A képletben csak az oszlop helye változik (nem állandó). Másoláskor a sor (ami rögzített) nem változik a képletben.

Pl. jobbra másoláskor =B4\*A\$1 → =C4\*B\$1

Lefele másoláskor: =B4\*A\$1 → =B5\*A\$1

#### 4. **Operátorok**

1. **Számítási operátorok:**

műveleti jelek: + - \* / % ^

2. **Összehasonlító operátorok:**

Két érték összehasonlítására szolgál. Az összehasonlítás eredménye, igaz vagy hamis logikai érték lesz.

egyenlő =

nagyobb >

kisebb <

nagyobb egyenlő >=

kisebb egyenlő <=

nem egyenlő <>

3. **Szövegösszeűző operátor:**

& két értéket kapcsol össze. Összeűzheti cellák tartalmát.

4. **Hivatkozási operátor:**

: tartomány operátor (kettőspont) – a tartomány első és utolsó tagját választja el.

Pl. =SZUM(A1:D9) – Az A1-es, D9-es és a köztük lévő összes cella értékeit összeadja.

; egyesítő operátor (pontos vessző) – a hivatkozás tagjait választja el a felsorolásban.

Pl. =SZUM(A1;D9) Csak az A1-es és D9-es cella értékeit adja össze.

5. **Szöveg megjelenítő operátor:**

”szöveg” (idéző jel). Az idéző jel közé írt szöveg megjeleníthető, akár egy képletben is.

6. **Abszolút hivatkozási operátor:**

\$ (dollár jel). Helyhez rögzíti az adott oszlopot, vagy sort, vagy mindkettőt a cella nevében. Másoláskor a rögzített sor, vagy oszlop nem változik. Gyorsbillentyűje az F4. Többször megnyomva használjuk.

## 5. Függvények

Előre elkészített képletek. Gyorsítják és könnyítik a munkát.

- **Jellemzői:**
  - Használatuk során meg kell adni, melyik értékekkel végezzük el a műveletet.
  - Az értékeket, a függvény zárójelen belüli alkotóit **argumentumnak** nevezzük.
  - A függvények felépítése, a függvényalkotók sorrendje meghatározott, azt nem módosíthatja a felhasználó.
  - A függvényekkel is végezhetőek műveletek. Egyik függvényből, például kivonhatjuk a másikat stb.
  - A függvények egymásba ágyazhatóak.
  - A függvények tartalma, értéktartománya módosítható. A cellában lévő függvény a szerkesztő lécben is megjelenik, nem csak a cellában. Mind a két helyen szerkeszthető.
- **Felépítése:**

=Függvény neve()
------------------

- = egyenlőségjellel kezdődik,
  - ezt követi a függvény neve,
  - az értékek, az argumentumok () íves zárójelek közé kerülnek.
- =SZUM()  
=SZUM(B2:D21)

### ARGUMENTUM NÉLKÜLI FÜGGVÉNYEK

Az argumentum nélküli függvényeknek a zárójelen belül nem kell értéket adni. Beírjuk egy cellába a függvény nevét pl. =MA() és készen is van a függvény.

- **MA függvény: =MA()**  
Az aktuális dátumot, dátumként formázva adja vissza.
- **PI függvény: =PI()**  
A  $\pi$  (3,14) értékét adja vissza 15 jegy pontossággal.
- **Vél függvény: =Vél()**  
Véletlen számot ad, ami 0, vagy nullánál nagyobb és 1-nél kisebb.

### EGY ARGUMENTUMOS FÜGGVÉNYEK

Az egy argumentumos függvényeknél a zárójelen belül meg kell adni, melyik cellára, cellákra vonatkozzon a számítás. Leggyakrabban a tartományt, tartományokat jelölünk ki!

- **SZUM függvény: =SZUM()**  
Összeadja a kiválasztott cellákban lévő a számokat.  
Pl. =SZUM(B2:D21), =MAX(H3:H11;J3:J11;H17;J17)
- **ÁTLAG függvény: =ÁTLAG()**  
Átlagolja a kiválasztott cellákban lévő számokat. Számtani közepüket veszi.  
Pl. =ÁTLAG(B2:D21), =ÁTLAG(B1:D4;B7:D11;F2;F7)

- MAX függvény: =MAX()**  
 Megjeleníti a kiválasztott cellák számai közül a legnagyobb számot. A logikai értékeket és a szöveget figyelmen kívül hagyja.  
 Pl. =MAX(E2:E11), =MAX(H3:H11;J3:J11;H17;J17)
- MIN függvény: =MIN()**  
 Megjeleníti a kiválasztott cellák számai közül a legkisebb számot. A logikai értékeket és a szöveget figyelmen kívül hagyja.  
 Pl. =MIN(E2:E11), =MIN(H3:H11;J3:J11;H17;J17)
- DARAB függvény: =DARAB()**  
 Megszámolja, hogy a kiválasztott cellákban mennyi darab számot tartalmazó cella van.  
 Pl. =DARAB(A2:A21), =DARAB(C8:C16;E8:E16;G8;G10;G12)
- DARABÜRES függvény: =DARAB()**  
 Megszámolja, hogy a kiválasztott cellákban mennyi üres cella van. Azaz mennyi darab olyan cella van, ami nem tartalmaz adatot. Adat lehet szám, vagy szöveg.  
 Pl. =DARABÜRES(C1:E14)

### **TÖBB ARGUMENTUMOS FÜGGVÉNYEK**

- A több argumentumos függvényeknél a zárójelen belül meg kell adni, melyik cellát, cellákat vegye figyelembe a függvény. Ezt még legalább egy argumentum követi, ami a függvény típusától függ, hogy mit tartalmaz.
- Kettőnél több argumentuma is lehet a függvénynek.
- Az argumentumokat pontosvessző választja el egymástól.

- DARABTELI függvény: =DARABTELI(tartomány;"kritérium")**

  - Megszámolja a feltételnek eleget tevő nem üres cellákat
  - Ez a függvény kettő argumentumot tartalmaz; tartomány és kritérium.
  - A kritérium lehet **szám, vagy szöveg**.

Pl. Mennyi darab 14-es szám van a tartományban? =DARABTELI(E1:E14;14)  
 Mennyi megfelelt szót tartalmaz a tartomány? =DARABTELI(G1:G14;"megfelelt")

Érdemjegyek száma	
5	=DARABTELI(\$I\$3:\$I\$23;M10)
4	DARABTELI(tartomány; kritérium) 4
3	2
2	12
1	2

**Hivatkozhatunk szöveget, vagy számot tartalmazó cellára is a feltétel megadásakor.**

A példán a függvény összeszámoltatja az 5-ös jegyek számát a tartományban. Feltételnek azonban nem az 5, hanem a számot tartalmazó cella neve szerepel, ami az M10.

### **FONTOS!**


A függvényben a tartomány (I\$3:I\$23 – a sorok) a képlet lefele másolása miatt lett rögzítve!  
 Mivel a többi érdemjegy darabszámát is meg kell számolni!  
 Ha rögzítjük a tartományt, a képlet lefele másolásával elvégeztük a feladatot.

**HA függvény: =HA(logikai\_vizsgalat;Érték\_ha\_igaz;Érték\_ha\_hamis)**

- Ellenőrzi a feltételt. Ha a feltétel igaz (teljesül), akkor az egyik értéket adja vissza. Ha a feltétel hamis (nem teljesül), akkor a másikat.
- Háromargumentumos függvény.
- A **logikai vizsgalat** során megadjuk a cellát, ami az értéket tartalmazza és megadjuk a feltételt.
- **Érték ha igaz.** Itt adjuk meg, mit jelenítsen meg a cellába, ha a logikai vizsgalat igaznak bizonyul.
- **Érték ha hamis.** Itt adjuk meg, mit jelenítsen meg a cellába, ha a logikai vizsgalat hamisnak bizonyul.
- Az igaz és a hamis érték is lehet szám, vagy szöveg, esetleg üresen hagyott cella.
- Üresen hagyott cellát akkor ad vissza igaz, vagy hamis értékként a táblázatkezelő, ha az **idézőjelek közé egy szóközt** írunk be.

Pl. Ha a megvizsgáltatott cella 20-nál nagyobb számot tartalmaz, akkor jelenítse meg a függvény a „megfelelt” szót, ellenkező esetben a „megbukott szót”

=HA(E1>20;"megfelelt";"megbukott")



Ha azt szeretnénk, hogy a megbukott szó helyett semmi se jelenjen meg a függvény a következő képen módosul: =HA(E8>20;"megfelelt";" ")

**A Ha függvény egyik fő jellemzője, hogy használata során, illeszthetünk bele más függvényeket, vagy számításokat is.**

Pl. Jelenítsen meg a H oszlopban képlet segítségével egy felkiáltójelet, ha a névhez tartozó pontszám az átlagpontnál nagyobb! Ellenkező esetben ne jelenjen meg semmi. A megoldás során alkalmazzon másolást!

=HA(G3>ÁTLAG(\$G\$3:\$G\$23);"!";" ")

Logikai vizsgalat alá véve a G3-as cella. Ha a benne lévő érték, nagyobb, mint az összes diák pontszáma alapján kiszámítható átlag, ami a G3:G23-as tartomány alapján számítunk ki egy átlag függvénnyel, G3>ÁTLAG(\$G\$3:\$G\$23) akkor.

Érték ha igaz: "!"

Érték ha hamis: " "

	A	G	H	I	J	K
	Név	Össz pontszám		Osztályzat	A határidő és a leadás közötti napok száma:	Ha az összpontszám 55 alatt van, PÓTLÁS
2						
3	Balogh Zsófia	79	=HA(G3>ÁTLAG(\$G\$3:\$G\$23);"!";" ")			
4	Nagykálózi Rozália	57	HA(logikai_vizsgalat; [érték_ha_igaz]; [érték_ha_hamis])			
5	Debreceni Timót	54		1	3	PÓTLÁS
6	Reider Rajmond	97	!	4	7	
7	Bakos Kinga	66		2	17	
8	Gál Edit	88	!	4	20	
9	Budaházy Gertrúd	69		2	18	
10	Bach Edit	92	!	4	2	

- **SZUMHA függvény: =SZUMHA(tartomány;"kritérium")**

**1. változat a használatra**

- Összeadja a kiválasztott cellákban lévő számokat, ha azok megfelelnek a megadott feltételnek. (Feltétel = kritérium)


- Ez a függvény **kettő argumentumot tartalmaz; tartomány és kritérium.**

- A zárójelben meg kell adni a tartományt és a feltételt.

- A kritériumot idéző jelek közé kell tenni.

Pl. Adja össze a tartományban a tíznél nagyobb számokat!

=SZUMHA(C1:D10;">10")



**2. változat a használatra**

- Összeadja a kiválasztott cellákban lévő számokat, ha azok megfelelnek a megadott feltételnek. (Feltétel = kritérium)

- Ez a függvény **három argumentumot tartalmaz; tartomány, kritérium, összeg tartomány.**


- **Tartomány:** a táblázatban a kiértékelendő cellatartomány, amiben a feltétel is van. Ha nem egy oszlopot adunk meg, akkor a kiértékelendő oszlopnak kell az első oszlopnak lennie a tartomány kijelölésekor.

- A **kritériumot** idéző jelek közé kell tenni, vagy egy cellára hivatkozva, a cella tartalmával kell megegyeznie a feltételnek.

- Az **összeztartomány** azok a cellák, amelyekből összeadásra kerülnek azok a számok, amik a feltételnek eleget tesznek.

Pl. Az alábbi táblázatban minden névhez tartozik egy kategória és egy költségtérítési összeg. Adja össze a tartományban egy külön táblázatban, hogy kategóriánként mennyi költségtérítés lett kifizetve!

=SZUMHA(B\$2:C\$7;E2;C\$2:C\$7)



INDEX		=SZUMHA(\$B\$2:\$C\$7;E2;\$C\$2:\$C\$7)				
	A	B	C	D	E	F
1	Név	Kategóriák	Költségtérítés		Költség összesítés kategóriánként	
2	Balogh Zsófia	A	1500		A	=SZUMHA(\$B\$2:\$C\$7;E2;\$C\$2:\$C\$7)
3	Nagykálózi Rozália	A	2000		B	3000
4	Debreceni Timót	B	2000		C	2000
5	Reider Rajmond	C	500			
6	Bakos Kinga	B	1000			
7	Gál Edit	C	1500			

**FONTOS!**

A függvényben a tartományok a lefele másolás miatt lettek rögzítve! Mivel B-t és C-t is ki kell számolni!

Ha rögzítjük a tartományokat, (sorokat) a képlet lefele másolásával elvégeztük a feladatot.

**ÁTLAGHA függvény:** =ÁTLAGHA(tartomány;kritérium;átlagtartomány)

- **Átlagolja** a kiválasztott cellákban lévő számokat, ha azok megfelelnek a megadott feltételnek.

- **tartomány:** a táblázatban a kiértékelendő cellatartomány, amiben a feltétel is van.


Ha nem egy oszlopot adunk meg, akkor a kiértékelendő oszlopnak kell az első oszlopnak lennie a tartomány kijelölésekor.

- **kritérium:** feltétel, ami alapján kiválasztjuk a táblázatból azokat az értékeket, amik a számításhoz szükségesek

- **átlagtartomány:** az oszlop, amiben az átlagolandó értékek vannak.

Pl. Számolja ki a férfiak átlagát minden tárgyra a 12. sorban a tantárgyak alatt! A férfiak kódja: 1.

= ÁTLAGHA(\$B2:\$H11;1;D2:D11)



	A	B	C	D	E	F	G	H
1	<b>név</b>	<b>nem</b>	<b>kor</b>	<b>angol</b>	<b>francia</b>	<b>számtech</b>	<b>protokoll</b>	<b>pénzügy</b>
2	Rigó Sarolta	2	17	45	60	55	51	88
3	Dobos Kálmánné	2	22	40	94	100	92	100
4	Petrus Katalin	2	23	70	74	65	98	63
5	Vágó Zoltán	1	23	60	72	65	74	70
6	Vörös R.	2	26	100	95	98	98	100
7	Füstös László	1	18	40	58	40	83	65
8	Lendvai Gáborné	2	19	80	98	90	93	100
9	Vassné H. Gabriella	2	21	80	90	22	87	54
10	Hámori Sándor	1	17	90	45	54	91	78
11	Kovács Judit	2	18	60	50	54	80	56
12	<b>Férfiak átlaga</b>			<b>63,3</b>	<b>58,3</b>	<b>53,0</b>	<b>82,7</b>	<b>71,0</b>
13								

### FONTOS!

A függvényben a tartományok a jobbra másolás miatt lettek rögzítve! Mivel a többi tantárgyra vonatkozólag is ki kell számolni az átlagokat!

Ha rögzítjük a tartományt (az oszlopokat), a képlet jobbra másolásával elvégeztük a feladatot.



## M4 – TÁBLÁZATKEZELÉS ALAPJAI


**FKERES függvény:** =FKERES(Keresési\_érték;Tábla;Oszlopszám)

- Egy megadott értéket megkeres, egy előre elkészített segédtablában. Eredményül megjeleníti a segédtablában az értékhez tartozó számot, vagy szöveget.
- **keresési érték:** az amihez értéket, vagy szöveget szeretnénk rendelni,
- **tábla:** az előre elkészített segéd tábla, - csak az adatok - az oszlopfejléc vagy sorfejléc nélkül,
- **oszlópszám:** az előre elkészített segéd tábla azon oszlopának a száma, amit meg szeretnénk jeleníteni.

Pl.

1. A táblázat C oszlopában, minden személyhez rendelve, meg szeretnénk jeleníteni a személy összpontszáma alapján, hogy milyen osztályzat jár neki. A ponthatárokat a segéd tábla tartalmazza.

=FKERES(B2;\$F\$2:\$G\$6;2) A sor eredménye: 4.



FKERES							
=FKERES(B2;\$F\$2:\$G\$6;2)							
	A	B	C	D	E	F	G
1	Név	össz pontszám	osztályzat	ajándék		Pontszám	Jegy
2	Balogh Zsófia	89	=FKERES(B2;\$F\$2:\$G\$6;2)			0	1
3	Nagykálózi Rozália	100	FKERES(keresési_érték; tábla; oszlop_száma; [tartományban_keres])				2
4	Debreceni Timót	79	3	naptár		72	3
5	Reider Rajmond	55	2	nincs		88	4
6	Bakos Kinga	69	2	nincs		99	5
7	Gál Edit	74	3	naptár			
8						Pontszám	ajándék
9						0	nincs
10						70	naptár
11						90	könyv

Pl.

2. A táblázat D oszlopában, minden személyhez rendelve, meg szeretnénk jeleníteni a személy összpontszáma alapján, hogy milyen ajándék jár neki. Az ajándékokat a segéd tábla tartalmazza.

=FKERES(B2;\$F\$9:\$G\$11;2) A sor eredménye: naptár.

FKERES								
=FKERES(B2;\$F\$9:\$G\$11;2)								
	A	B	C	D	E	F	G	H
1	Név	össz pontszám	osztályzat	ajándék		Pontszám	Jegy	
2	Balogh Zsófia	89		4	=FKERES(B2;\$F\$9:\$G\$11;2)			1
3	Nagykálózi Rozália	100		5	FKERES(keresési_érték; tábla; oszlop_száma; [tartományban_keres])			
4	Debreceni Timót	79		3	naptár		72	3
5	Reider Rajmond	55		2	nincs		88	4
6	Bakos Kinga	69		2	nincs		99	5
7	Gál Edit	74		3	naptár			
8						Pontszám	ajándék	
9						0	nincs	
10						70	naptár	
11						90	könyv	

**HOL.VAN függvény: = HOL.VAN (keresési\_érték;tábla;egyezési\_típus)**

- Egy elemnek az oszlopban elfoglalt relatív helyét adja vissza. **Megadja melyik sorban található az elem.** Tehát az elem helyét és nem magát az elemet adja vissza.
- Relatív a hely, a rendezés miatt. Ha átrendezzük a táblázatot, az elem helye megváltozhat.
- Háromargumentumos függvény.
- **Keresési érték:** lehet szám, vagy szöveg.
- **Tábla:** amiben keresünk. Itt a tábla adott oszlopának tartományát kell megadni. (Nem az egész táblát, csak az oszlop tartományát. Pl. C1:C1000)
- **Egyezési típus:**

- a. Lehetőséget ad *keresési feltétel megadására*. Melyik szám értékének a helyét keressük az *általunk megadott számhoz viszonyítva*. Annál nagyobbat? Azzal egyenlőt? Annál kisebbet?
- b. Az egyezési típust számokkal jelöljük.  
Ezek a számok lehetnek: 1 0 -1  
A számokhoz rendezési feltétek is tartozhatnak. Ha a számhoz tartozó rendezést nem végezzük el, hibás eredményt kapunk.
- c. **1 szám jelentése:**  
A **legnagyobb** értéket keresi, **ami kisebb, vagy egyenlő**, mint a keresési érték. A táblának **emelkedő** sorrendben rendezettnek kell lennie.  
Pl. Melyik sortól kezdődik az 1450-et leginkább közelítő, egyenlő, vagy annál kisebb úrtartalmú autók listája egy olyan táblázatban, ahol a D oszlop D1:D1000 tartományában tároljuk a gépjárművek úrtartalmát.  
=HOL.VAN(1450;D1:D1000;1) rendezés növekvő
- d. **0 szám jelentése:**  
Az általunk megadott keresési értékkel **pontosan egyezőnek** a helyét adja vissza.  
A táblának nem kell rendezettnek lennie. **Rendezetlen táblában** is használható. **Szöveges értéket is megadhatunk!**  
Pl. Melyik sorban található az első mercedes típusú gépjármű egy olyan táblázatban, ahol a C oszlop C1:C1000 tartományában tároljuk a gépjárművek típusát.  
=HOL.VAN("Mercedes";C1:C1000;0) nem kell rendezni
- e. **-1 szám jelentése:**  
A **legkisebb** értéket keresi, **ami nagyobb, vagy egyenlő**, mint a keresési érték. A táblának **csökkenő** sorrendben rendezettnek kell lennie.  
Pl. Melyik sortól kezdődik az 1250-et leginkább közelítő, egyenlő, vagy annál nagyobb úrtartalmú autók listája egy olyan táblázatban, ahol a D oszlop D1:D1000 tartományában tároljuk a gépjárművek úrtartalmát.  
=HOL.VAN(1250;D1:D1000;-1) rendezés csökkenő

Emlékeztető tábla:

1	↑	legnagyobb ÉRTÉK	<=	KERESÉSI ÉRTÉK
0	●	egyező ÉRTÉK	=	KERESÉSI ÉRTÉK
-1	↓	legkisebb ÉRTÉK	>=	KERESÉSI ÉRTÉK

## M4 – TÁBLÁZATKEZELÉS ALAPJAI

Pl. A táblázat melyik sorában található az adott rendszám BHO336?

- **Keresési érték:** BHO336

- **Tábla:** A tábla adott oszlopának tartományát kell megadni. Ez itt a B1:B1001. Ha a fejléct is bele vesszük a jelölésbe, ami az első sor, akkor a munkalap sorával meg fog egyezni a visszaadott érték. Ha a fejléct nem vesszük bele, B2:B1001 akkor nem fog megegyezni a munkalap sorával. Ezért is relatív ez a függvény.

- **Egyezési típus:** 0 Mert az általunk megadott keresési értékkel pontosan egyezőnek a helyét adja vissza.

=HOL.VAN("BHO336";B1:B1001;0)

I3		fx		=HOL.VAN("BHO336";\$B\$1:\$B\$1001;0)				
	A	B	I	J	K	L	M	
1	<b>Kód</b>	<b>Rendszám</b>						
2	EF306044	SUC581	<b>Melyik sorban található BHO336 rendszám?</b>					
3	BA312699	FZA390	336					
4	AA827005	STD181	<b>Melyik sorban található az AB47263 kód?</b>					
5	AB100715	LGP936	62					
6	AA831663	ULK754						
7	BA915456	ZFR127						
8	EF623738	LTD981						
9	GH440100	UVB190						
10	AB624376	BBV909						
11	AB503259	VZA890						
12	BA228220	DZA190						
13	AB290951	NSE372						

A **HOL.VAN** függvényt általában nem egyedül szoktuk használni. Gyakran illesztünk bele pl. minimum vagy maximum függvényt. Illetve a **HOL.VAN** függvényt másik függvényben - az **INDEX** függvénybe illesztjük bele.

Pl. A táblázat melyik sorában található az legmagasabb pontszám:

=HOL.VAN(MAX(B2:B7);B2:B7;0)

- **Keresési érték:** MAX(B2:B7) a maximum függvénnyel a legmagasabb szám.

- **Tábla:** Ez itt a B2:B7 oszlop. Itt a fejléct nem vesszük bele, azért mert a maximum függvény is a 2. sortól kezdődik.

- **Egyezési típus:** 0

D3		fx		=HOL.VAN(MAX(B2:B7);B2:B7;0)					
	A	B	C	D	E	F	G	H	
1	<b>Név</b>	<b>össz pontszám</b>							
2	Balogh Zsófia	89		A legmagasabb pontszám helye a táblázatban:					
3	Nagykálózi Rozália	100		2					
4	Debreceni Timót	79							
5	Reider Rajmond	55							
6	Bakos Kinga	69							
7	Gál Edit	74							

- **INDEX függvény: =INDEX(tömb;sor\_ száma;oszlop\_ száma)**
  - Érték, vagy értékre történő hivatkozást szolgáltat egy táblázatból.
  - Megjeleníti a táblázat adott sorába, és adott oszlopában lévő értéket – a cella tartalmát.
  - Háromargumentumos függvény.
  - **Tömb:** egységes, egybefüggő tartomány, amiben az értéket keressük. Az egész táblázat adattartományát célszerű kijelölni. A fejléct általában nem jelöljük bele.
  - **Sor száma;** a kijelölt tömbben annak a sornak a száma, amelyben lévő értéket keressük.
  - **Oszlop száma;** a kijelölt tömbben annak az oszlopnak a száma, amelyben lévő értéket keressük. A betűjelek helyett számozzuk a tömb oszlopait.

Pl. Milyen érték található a munkalap B537-es cellájában?

=INDEX(A1:C1001;537;2)

- **Tömb:** A1:C1001 Itt a fejléc is, mert a munkalap sorát adja meg a feladat és nem a táblázatét.

- **Sor száma;** 537

- **Oszlop száma;** 2 Mert a kijelölésnek a B oszlop a 2. oszlopa.

	A	B	C	H	I	J
1	<b>Kód</b>	<b>Rendszám</b>	<b>Típus</b>			
2	EF306044	SUC581	Audi	Érték a B 537-es cellába:		JKL345
3	AB183339	UOH763	Subaru			
4	BA261786	DOH563	Skoda			
5	AB818202	LFR527	Subaru			
6	EF123224	DZA390	Fiat			
7	AA37017	SRF372	Skoda			
8	EF413451	PBV909	Skoda			

Az **INDEX** függvényt általában nem egyedül szoktuk használni. Gyakran illesztünk bele más függvényt. Pl. a **HOL.VAN** függvényt.

Pl. A D2-es cellába, ha beírunk a táblázatból egy nevet, a tőle jobbra lévő cellába képlet segítségével írassa ki, hogy melyik megyéből jött az illető.

1. lépés; a név oszlop hányadik sorában szerepel a név. (hol.van függvény)

2. lépés; az előző függvényt illesszük be az index függvény második argumentumaként, ahol a sorok számát kell megadni.

=INDEX(A2:B7;HOL.VAN(D2;A2:A7;0);2)

	A	B	C	D	E
1	<b>Név</b>	<b>Megye</b>		<b>Név</b>	<b>Megye</b>
2	Balogh Zsófia	Győr-Moson-Sopron		Bakos Kinga	Fejér
3	Nagykálózi Rozália	Veszprém			
4	Debreceni Timót	Fejér			
5	Reider Rajmond	Veszprém			
6	Bakos Kinga	Fejér			
7	Gál Edit	Győr-Moson-Sopron			

**FÜGGVÉNYEK EGYMÁSBA ÁGYAZÁSA**

Bizonyos esetekben szükség lehet arra, hogy az egyik függvényt a másik függvény egyik argumentumaként használjuk.

Pl. A következő képlet beágyazott ÁTLAG függvényt használ, és az eredményt az 50 értékkel hasonlítja össze.

Ha a feltétel igaz, => az F2:F5 tartomány átlaga nagyobb, mint 50, akkor összeadja a G2:G5-ös tartományt. Itt beágyazott SZUM függvényt használ.

Ha a feltétel hamis, => az F2:F5 tartomány átlaga egyenlő, vagy kisebb mint 50, akkor 0-t ír ki.

$$\text{=HA(ÁTLAG(F2:F5)>50;SZUM(G2:G5);0)}$$

Pl. Az F4-es cellába írjon képletet, amely annak a megyének, illetve városnak a nevét jeleníti meg, ahol a legalacsonyabb volt a munkanélküliek száma!

=INDEX(A2:B14;HOL.VAN(MIN(B2:B13);B2:B13;0);1)

1. lépés: A minimum megkeresetése függvénnyel **MIN(B2:B13)**

2. lépés: A minimum függvény beillesztése a hol.van függvény keresési érték helyére.  
**HOL.VAN(MIN(B2:B13);B2:B13;0)**

3. lépés: A hol.van függvény beleillesztése az index függvénybe, a sor szám helyére.  
**=INDEX(A2:B14;HOL.VAN(MIN(B2:B13);B2:B13;0);1)**

	A	B	C	D	E	F	G
1	<b>Regisztrált munkanélküliek száma, 2003 júliusában</b>						
2	Vas	5 842	2,9%				
3	Zala	9 197	4,6%				
4	Tolna	10 885	5,4%	Ahol a legalacsonyabb:		Vas	
5	Nógrád	12 179	6,0%	Ahol a legmagasabb:		BAZ	
6	Heves	12 374	6,1%				
7	Fejér	13 963	6,9%				
8	Csongrád	14 685	7,3%				
9	Békés	16 210	8,0%				
10	Pest	16 741	8,3%				
11	Baranya	17 854	8,8%				
12	Budapest	18 767	9,3%				
13	BAZ	53 427	26,4%				
14	Átlag	16 843,7					
15	Összesen	202 124,0					

- **Érvényes visszaadott értékek:**

Amikor beágyazott függvényt argumentumként használunk, annak ugyanolyan típusú értéket kell adnia, amelyet az argumentum használ.

Ha például az argumentum IGAZ vagy HAMIS értéket ad, a beágyazott függvénynek is IGAZ vagy HAMIS értéket kell visszaadnia. Ha nem ezt teszi, a Microsoft Excel az #ÉRTÉK! hibaértéket jeleníti meg.

- **A beágyazási szint határértéke:**

Egy képlet legfeljebb hét egymásba ágyazott függvényszintet tartalmazhat.

Amikor a B függvényt az A függvény argumentumaként használjuk, akkor a B függvény második szintű függvény.

Például az ÁTLAG és a SZUM függvény második szintű, mivel mindkettő a HA függvény argumentuma. Az ÁTLAG függvénybe ágyazott újabb függvény harmadik szintű függvény lenne, és így tovább.